Tutorial 30: Complément sur les matrices

Nombres décimaux dans une matrice et affichage des éléments de la matrice

Introduction

Bon, Ian a étendu ces connaissances sur les matrices et les listes… et je vais vous en faire part! Cela réutilise le premier tutorial sur les matrices et le tutorial sur les nombres décimaux - rien de mieux pour la pratique. Ce tutorial est égtalement composé de deux programmes. Le premier pour mettre deux nombres décimaux dans une matrice, le deuxième pour afficher les éléments de la matrice.

Nouvelles commandes

_adrmele

- calcule l'adresse d'un élément de la matrice

_creatermat
- crée une matrice aux dimension hl

_getmtoop1
- prend un élément et le stocke dans Op1

Le code

Voici le code…

.NOLIST

#define equ .equ

#define EQU .equ

#define end .end

#include "ti83asm.inc"

#include "tokens.inc"

.LIST

_creatermat .equ 446Ah

rmatobj .equ 02h

.org 9327h

     call _zerooop1    ; clear op1

     ld hl,op1         ; copy var name for matrix [A] into op1

     ld (hl),rmatobj   ;

     inc hl            ;

     ld (hl),tvarmat   ;

     inc hl            ;

     ld (hl),tmata     ;

     call _chkfindsym  ; look it up

     call nc,_delvar   ; if it is there delete it

     ld hl,$0102       ; 1 by 1

     call _creatermat  ; create it

     push de                ; save matrix address

     ld bc,$0101            ; element 1

     call _adrmele          ; calculate element address

     ex de,hl               ; de -> element address

     ld hl,element1         ; hl -> num

     ld bc,9                ; 9 bytes

     ldir                   ; copy num to element

     pop de                 ; restore matrix address

     ld bc,$0102            ; element 2

     call _adrmele          ; calculate element address

     ex de,hl               ; de -> element address

     ld hl,element2         ; hl -> num

     ld bc,9                ; 9 bytes

     ldir                   ; copy num to element

     ret

element1:

     .db $00,$80,$75,$00,$00,$00,$00,$00,$00  ; 7.5

element2:

     .db $80,$82,$73,$45,$67,$23,$89,$00,$00  ; -734.5672389

.end

END

Relativement simple quand vous le voyez non? Maintenant vous devez être capable de créer, détecter les matrices et de stocker des éléments dedans. C'est utile si vous créer des éditeurs de niveaux pour vos jeux, les matricfes sont très pratiques pour stocker les niveaux. Maintenant, afficher les éléments de la matrice.

.NOLIST

#define equ .equ

#define EQU .equ

#define end .end

#include "ti83asm.inc"

#include "tokens.inc"

.LIST

_errundefined .equ 467Bh

rmatobj .equ 02h

.org 9327h

     call _zerooop1      ; clear op1

     ld hl,op1           ; copy var name for matrix [A] into op1

     ld (hl),rmatobj     ;

     inc hl              ;

     ld (hl),tvarmat     ;

     inc hl              ;

     ld (hl),tmata       ;

     call _chkfindsym    ; look it up

     jp c,_errundefined  ; error if not found

     push de             ; de -> matrix address

     pop hl              ;

     call _ldhlind       ; hl -> matrix size

     ld bc,$0101         ; bc -> element number

rowl:

     push hl             ; store count

     push bc             ; store element num

coll:

     push hl             ; store count

     push bc             ; store element num

     push de             ; store matrix address

     call _getmtoop1     ; get element to op1

     ld a,16             ; format number for display

     call _formreal      ;

     ld hl,16            ;

     sbc hl,bc           ;

     ld a,l              ;

     ld (curcol),a       ;

     ld hl,op3           ;

     call _puts          ; display it

     pop de              ; restore matrix address

     pop bc              ; restore element num

     inc c               ; increment column

     pop hl              ; store count

     dec l               ; decrement column count

     ld a,l              ; see if column count is at 0

     or a                ;

     jr nz,coll          ;

     pop bc              ; restore element num

     inc b               ; increment row

     pop hl              ; restore count

     dec h               ; decrement row count

     ld a,h              ; see if row count is at 0

     or a                ;

     jr nz,rowl          ;

     ret

.end

END

He-he, ce programme demande un petit peu de réflexion. Si vous n'y arrivez pas, faites un petit diagramme… tout ce que Ian fait c'est qu'il prend les variables et fait des sauts dans la matrice. La routine est relativement identique à la précédente.

Conclusion

Bon, enfin, les matrices sont totalement étudiées! Place aux listes…

Merci à Ian Graf

